
CHAPTER 4
Delinquency, Default, and

Loss Analysis

N
early every asset class in structured finance is subject to loan obligors not paying.
This eventually generates losses. While the concept of an obligor defaulting

and causing a loss is not difficult to understand, historical loss reporting, future
loss projection, and how a structured transaction model is constructed to handle
these are often a cause of confusion. Because there is no standardized method
for reporting, projecting, or modeling loss, data is often presented in inconsistent
and sometimes misleading formats. The most favored format of understanding
historical loss information is the static loss report, which tracks losses for each
origination period over time. This format is ideal for building historical loss curves
and producing a projected loss curve with severity and timing that is predicated on
prior asset performance.

Incorporating loss into a model can also be done a number of ways depending
on certain rating agency methodologies and subjective formula techniques. This
chapter lays the groundwork for understanding loss and shows one possible method
for integrating the concept into a functioning model. Later in this book there are
discussions about other methodologies.

DELINQUENCIES VERSUS DEFAULTS VERSUS LOSS

One reason understanding loss can be so confusing is the terminology that is involved
in the process leading up to a loss. To be clear prior to any use, the following related
terms are defined:

■ Delinquency. When a payment is due on a specific date and the obligor fails
to pay the payment by that date the obligor is delinquent. A more granular
analysis of delinquency typically splits delinquency into monthly intervals such
as 1 to 30 days delinquent, 31 to 60 days delinquent, 61 to 90 days delinquent,
and so on. A key point to an obligor being considered delinquent is that there is
an expectation of more payments and the possibility that the obligor will repay
enough to no longer be considered delinquent.

■ Default. Depending on company and legal definitions, a delinquency eventually
becomes a default. When obligors are considered defaulted, they are no longer
expected to make regular payments and legal action typically begins. It is very
important to realize that the timeframe it takes for an obligor to transition from
delinquency to default is usually set by legal definitions in structured transactions
and that loss expectations can change depending on the time it takes to classify
an obligor as a default.

■ Loss. When a loan is considered a default, some companies instantly write
the principal balance of the loan off. This is known as a gross loss. As time
progresses, sometimes a recovery is made either through the sale of a repossessed
asset or continued pursuance of credit collection.

The well-known fixed income expert, Frank J. Fabozzi, stresses the importance
of terminology by defining a loss as follows:

A default is essentially a loan that does not get paid back. A loan may
go into the 90-day delinquent category or even into foreclosure, and still
be made current again. So only those loans that are ultimately liquidated
for non-payment are classified as defaulted loans. Losses refer to the dollar
amounts lost on defaulted loans.1

A useful example is to examine one loan in a hypothetical loss situation. Imagine
a mortgage loan that had an original balance of $1,000.00, a current balance of
$586.50, is in its 20th month and stops paying. Figure 4.1 depicts the timeline of
events and classifications that the loan would go through.

As can be seen from Figure 4.1, the loan is considered delinquent until 90 days
have passed by and then it becomes classified as a default. In this case, it is written off
instantly and the legal process begins. After 14 months, a court orders a foreclosure.
The asset takes three months to sell and an asset recovery is realized.

THE IMPORTANCE OF ANALYZING DELINQUENCY

Understanding the asset pool’s tendency to initially stop paying on time is a logical
starting point for a loss analysis. This is done by examining historical delinquency
data on similar assets and is important for two reasons: credit trends and liquidity.
Credit trends are an indication of whether assets will perform better or worse and
whether a greater or lesser amount of loss should be expected in the future. Liquidity
is a transaction-specific consideration that refers to the amount of actual cash in any
given period.

1Frank J. Fabozzi, Bond Credit Analysis: Framework and Case Studies (New Hope, PA: Frank
J. Fabozzi Associates, 2001), 245.

FIGURE 4.1 A timeline of the events that take place when an asset defaults.

The first concern, credit trends, is important because it could mean that a future
loss projection might have to be increased. Credit trends are noticeable when
a delinquency analysis is performed on existing assets. If, over the course of
origination, assets progressively have increased delinquency, the loss expectation
may be higher. Greater delinquency sometimes means that the company originating
the assets is chasing riskier assets, possibly for greater returns. In this case, the
obligors’ credit scores and interest rates should be examined more closely.

The second concern, liquidity, is extremely important in building a structure
that is sized and risk rated correctly. If assets consistently have a high delinquency,
this needs to be accounted for in a model; otherwise there might not be enough
cash to cover loss, principal, interest, and possibly fees. Delinquency directly affects
cash flow because a delinquent loan does not have any cash flow; but it is also not
considered a default. This means that its balance is not written down, excess spread
does not cover the loss, and no cash is received. After the liability side of the model
is complete, this concept becomes clearer.

Both concerns require the creation of historical delinquency curves. These curves
are useful in determining credit trends and the severity of delinquency. The curves
are not actually used in Project Model Builder, but they are created using a Model
Builder exercise since the analysis is very important.

MODEL BUILDER 4.1: BUILDING HISTORICAL DELINQUENCY CURVES

1. Ideally a historical delinquency analysis begins with data from assets similar to
those in the transaction. In this exercise, data has been provided in the Ch04
folder on the CD-ROM. Open the file MB4-1 Raw Data to access the raw
data. Save the file as Historical Delinquency Curves.xls.

2. There are two sections of data provided: The top section is the current balance of
loans, grouped by vintage, reported on a periods-out basis. The bottom section
contains the current balance of loans that are in the 30-day delinquency bucket
(that is, they have missed their payment date by 1 to 30 days). The best way to
understand the magnitude and trend of the delinquencies is to calculate them
on a percentage basis. First, to create an area for the rates, copy C37:AA37 and
paste this cell range in C66:AA66. Also copy B38:B62 and paste this cell range
in B67:B91. Labels for the axis of the ranges can also be copied down, but make
sure to leave row 65 blank for now.

3. The proper delinquency calculation is the current month’s delinquency over the
prior month’s ending balance. The formula for this is a simple division equation;
but it needs some cleaning up to prevent #DIV/0 errors. In cell C68 enter:

=IF(C39="","",C39/C7)

This formula cleans up zero-balance periods and divides the delinquency by the
current balance of the prior period.

4. To get a sense for all of the vintages, a weighted average is useful. As seen
in Chapter 3, calculating a weighted average is most simply done using a
SUMPRODUCT-SUM combination and is discussed in more detail in the
Toolbox section of that chapter. Also, to prevent picking up blank cells, an
OFFSET function is needed. Prior to creating the formula, the OFFSET needs a
reference value to determine how many cells of data each vintage contains. Since
January 2004 contains the maximum periods of data (24) and each successive
month has one less period of data, a column of decreasing values is necessary.
Starting in A68, enter 24, and then in B68 enter 23. Highlight both of these cells
and drag the decreasing value down to A91. The worksheet should look like
Figure 4.2 at this point.

5. Cell AC68 will contain the weighted average of the first period, AC69 the second
period, and so on. In cell AC68 enter:

=SUMPRODUCT(C68:OFFSET(B68,0,A68),C7:

OFFSET(B7,0,A68))/SUM(C7:OFFSET(B7,0,A68))

This weighted average formula multiplies each delinquency rate by the respective
vintage’s prior period current balance. This creates the weighting and then
divides the sum of those values by the sum of the balances to get the weighted

FIGURE 4.2 A standard set of delinquency curves organized by periods out.

FIGURE 4.3 Delinquency performance can be quickly
understood from a chart.

average rate. An OFFSET is used to make sure that no blank values are
picked up.

6. Charts are often seen to represent this data as seen in Figure 4.3.

The delinquency data in Model Builder 4.1 is useful because it shows the amount
that is delinquent 1 to 30 days. This is useful knowledge for later if a liquidity reserve



64 MODELING STRUCTURED FINANCE CASH FLOWS WITH MICROSOFT EXCEL

needs to be sized in the modeling. Also, notice that there is a slight trend in the
data. Earlier vintages have experienced slightly more delinquency than later vintages,
suggesting an improving trend in credit quality. This could possibly be an argument
for reducing loss projections.

Keep in mind that this is the most basic delinquency analysis that can be
completed. Often the process is repeated for each delinquency interval (i.e., 1 to
30 days, 31 to 60 days, and so on). Delinquency information can also be used to
create transition matrices, which can be used as a proxy for loss if there is a limited
amount of historical data. However, if there is static vintage data, historical loss
curves should be created and analyzed.

DERIVING HISTORICAL LOSS CURVES

Delinquencies that eventually pay only create liquidity issues and negative drag on
the transaction and are not as important as the loans that completely stop paying
and are considered defaults. Similar to delinquency, default expectations for a pool
of assets can be assumed from historical data. In fact, the analysis is very similar,
using a static type analysis.

The importance of a static type analysis should not be understated. To get a
comprehensive understanding of the performance of an asset it must be tracked
from its origination to maturity, without the influence of other loans from different
origination periods. This is the core concept behind a static analysis. The term static
is used because individual vintages are examined independent of other vintages. The
losses for all loans in a vintage are tracked and recorded as time progresses. The
comparison between loss and balance is only within the vintage under analysis. No
other loan’s losses or balances are added at any time. For example, in Figure 4.4,
originations are displayed across row 7. The date of these originations is directly
above in row 6. For each origination period the losses can be seen going vertically
in their respective columns.

A quick example of interpreting Figure 4.4 is looking at loans originated in
March 2004. In that month, there were $272 in losses four periods from origination.
The four periods out corresponds to July 2004, which is cell E11 in Figure 4.4.

It is important to fully understand what the $272 represents. In this example,
that amount is the gross loss reported in that month. Gross loss is the dollar
amount of loan balances that are assumed to remain unpaid. The typical method of
understanding the magnitude of the loss per period is by representing it as a fraction
of origination. Specifically, dividing the historical loss per period associated with a
specific origination by the origination amount results in a percentage of loss that the
loans in the origination pool experienced. Taking the earlier example of the $272 in
July 2004, divide the loss amount by the originations for March 2004 (i.e., 29,266).
This results in 0.93 percent as seen in Figure 4.5.

The rationale for presenting the loss data this way is that it allows for a direct
comparison of losses between origination pools. Also, this method of reporting

FIGURE 4.4 Raw loss data is best presented on a static loss basis.

FIGURE 4.5 The raw loss data should be converted to percentages of original
vintage balance.

prevents inaccurate loss percentages that could occur if the loss percentage was
calculated in a dynamic manner. Standard & Poor’s provides excellent reasoning on
why static loss data is preferred:

An analysis of static pool data is preferred, since it demonstrates loss
performance over the full liquidation period of a pool. Static pool data also
can provide insight into changing portfolio characteristics, underwriting, or
collection policies. Movement in such variables may not be readily apparent
in dynamic portfolio data because of the constantly changing mix in new and
aged receivables. Additionally, data based on annual losses can understate

loss performance during periods of rapid portfolio growth. When analyzing
high-growth portfolios, the current period’s losses may be compared against
the prior year’s portfolio. This ‘‘growth adjustment’’ takes into account
that losses on new originations may not happen immediately and, therefore,
occur in the next annual period.2

Tracking the loss history of loans based on origination gives a better estimation
of the asset performance over time.

Once the monthly loss amount is determined, it is very easy to calculate the
cumulative loss percentage, which is the sum of all the monthly loss percentages, up
to the current period. Using the same origination pool as the other examples (March
2004) it is evident that the cumulative loss through March 2006 is 12.99 percent
(see Figure 4.6).

FIGURE 4.6 A cumulative loss curve
sums the periodic loss amounts.

2Standard & Poor’s, Structured Finance: Equipment Leasing Criteria (New York:
McGraw-Hill, 1999), 17.

FIGURE 4.7 The difference between organizing data using a
periods-out versus a dates methodology.

As this point, it is worth noting that there can be some confusion as to how
the loss curves progress over time. One part to be mindful of is whether the unit
of time is represented as a period of time or a reference date. The method, used
for the examples thus far, represents time progression using periods and is called
the periods-out method. This method means that loans originated in March 2004
had .93 percent of loss four periods out (July 2004); .79 percent of loss five periods
out (August 2004); and so on. However, loss data can also be provided without the
periods but with dates instead. The dates typically start with the earliest origination
period and continue vertically on a monthly basis. The key difference is that the
dates method sets the time reference for all of the origination periods, while the
periods-out method has an independent time reference for each origination period.

In the dates method loans originated in March 2004 show $272 of loss in July
2004, seven cells down. The same origination period in the periods-out method
shows the $272 of loss four cells down. Essentially, in a periods-out method each
time progression of a cell is based off each origination pool date, while a ‘‘dates’’
method is based off of the very first origination date. To help make this clear, look
at Figure 4.7.

MODEL BUILDER 4.2: BUILDING HISTORICAL AND

PROJECTED LOSS CURVES

1. Open the file MB4-2 Raw Data.xls from the Ch04 folder on the CD-ROM.
This file is similar to the delinquency one, but with changes relevant to loss
analysis. The first noticeable difference is that there are only origination balances
for each of the pools. This is because static loss analysis is done off of the
original balance of the loan. Save this file under a different name so it can be
worked in.

2. Similar to the delinquency analysis, copy the origination dates and balances
to rows 37 and 38. Also copy the periods in column B so that it starts with
Origination Amt in row 38.

3. Starting in cell C39 enter the following formula:

=IF(C8="","",C8/C$38)

This is similar to the delinquency formula; but it bases all of the loss on original
balance instead of current balance. Copy this formula over the range C39:AA62.

4. So far monthly loss percentages have been created. This data is better analyzed
looking at cumulative percentages to see the ‘‘curve.’’ Copy the vintage dates
and origination amounts to row 65 and 66. Also copy the periods in column B
to start at row 66. The cumulative percentage is the current period’s loss plus
the previous period’s cumulative percentage. Before entering a formula, insert
a blank row in row 67. This should put a row of empty cells between the first
period and the origination amounts so the formula that will be entered does
not add the origination amounts to the cumulative loss percentage. With that
complete enter the following formula in C68:

=IF(C39="","",C39+C67)

Copy this formula over the range C68:AA91. The cumulative curves are often
summarized in a line graph that looks like Figure 4.8.

Before continuing further with Model Builder 4.2, it is important to understand
how to analyze the curves to use them for projections.

FIGURE 4.8 A graph of cumulative loss curves.

ANALYZING HISTORICAL LOSS CURVES

Vintage loss curves created using a static methodology have two important char-
acteristics that should be identified: severity and timing. The severity is the final
cumulative loss percent per vintage. This is how much of the original balance of a
particular vintage is assumed to be defaulted and uncollectible. The timing is how
much loss has been taken by a certain point in time, ending at the final maturity
of the assets. If the assets in the Model Builder example had final maturities of 24
months, then the timing of loss for any period can be determined by dividing the
cumulative loss percentage in that period by the final cumulative loss percentage
(period 24).

Loss timing is important to understand because it can have profound effects
on structured transactions. If the loss timing is front loaded, which means that
losses take place quickly the assets will erode quickly. This directly impacts excess
spread in a transaction, which is the first source of protection against loss. A
transaction modeled with a front-loaded curve versus a regular curve will require
more enhancement since there is less time for excess spread to generate. Back-loaded
curves, where losses take place near the end of the tenor of the assets also have
special effects on structured transactions. If loss does not take place until late
in the transaction, enhancement needs to be sized and kept for those periods.
If a transaction was modeled with a regular loss curve and losses were actually
back-loaded, important structural features such as triggers and reserve accounts
might be inadequate to protect against the back-loaded loss.

MODEL BUILDER 4.2 CONTINUED

1. Label cell AC38 Weighted Avg Curve. To get a summary of the severity of the
historical loss curves a weighted average curve needs to be created. This is done
using the following formula starting in AC39:

=SUMPRODUCT(C39:OFFSET(B39,0,A39),$C$38:

OFFSET($B$38,0,A39))/SUM($C$38:OFFSET($B$38,0,A39))

Copy this formula down to AC62. Also, since these are the monthly losses, sum
them up in AC64 to get the weighted average loss.

2. Timing should be analyzed on a monthly basis first and then cumulative. Take
the first period’s monthly loss amount and divide it by the sum of all the monthly
loss. Label cell AD38 Timing, and start the following formula in AD39:

=AC39/$AC$64

Copy this formula down to AD62. A sum of this column in AD64 should equal
100 percent.

3. A useful way to observe the timing is to make a cumulative timing curve. Do
this by entering the following formula in AE40:

=AE39+AD40

Notice that this started one more row down than the other formulas to avoid
having the label added in a formula. Copy this formula down to AE62. Model
Builder 4.2 will finish up after the next section on projecting loss curves.

PROJECTING LOSS CURVES

If no trend is evident and there are years of data that encompass the tenor of the
asset, then the weighted average curve created in the previous section can be used
as a projected loss curve. However, most of the time industries and companies go
through cycles of increasing and decreasing loss. Also, particularly with assets in
emerging markets, a relatively short time span of data is available. Both of these
issues create the need to project out loss curves.

The first issue, trending, is observed by looking at the same period for each
vintage. In Model Builder 4.2, the monthly losses have a noticeable decreasing trend.
Look at period 5 in Figure 4.9 and notice that in general each successive vintage after
January 2004 has a decreasing loss amount. Most of the periods are experiencing
such a trend. The company could argue that a weighted average curve based solely
on the data overstates loss because the newer vintages are expected to have a lower
loss amount in later periods, but these amounts are not reported and therefore not
captured in the weighted average loss curve.

A thorough loss analysis when trending is involved requires the ability to
observe the full spectrum of loss an asset may experience from origination to
maturity. Taking the weighted average losses for each period will only produce
accurate curves depending on the breadth of the historical loss data vis-à-vis the
age of the assets. The usefulness of the historical loss curves can be assessed by
determining how many of the loss curves have tracked data from origination to
maturity. As an example, assume the current date is January 2006 and in our
examples the data is provided as early as January 2004. Also assume that the final
maturity of the assets is 24 periods. This means that if originations and loss data is

FIGURE 4.9 Trends should be looked for in vintages across periods.

provided monthly, there could be one vintage that has reached maturity or ‘‘termed
out.’’ For instance, loans originated in January 2004, with a final maturity of 24
months should have all matured by January 2006. Since the loss data is from January
2004 through January 2006, there is loss history from every part of the loans’ term.
However, loans originated in April 2004 will only have a partial loss history, since
there would only be 21 months of data (May 2004 to January 2006).

If there is a trend in the data and there are few vintages that have ‘‘termed
out,’’ the earlier vintages will have a strong impact on the weighted average curve.
To account for such trends, the newer vintages need to be adjusted. For instance, if
losses are trending upwards and the later vintages aren’t ‘‘grossed up’’ for expected
loss, the weighted average method will understate loss. The opposite will occur if
losses are trending downwards, resulting in an overstatement of loss.

To account for trends, later vintages need to be adjusted using a timing curve
extrapolated from a set of ‘‘base’’ originations. A ‘‘base’’ origination should be a
historical origination from the static loss data that is demonstrative of the expected
timing of the assets. As long as the asset performance is not extremely volatile, it
would be logical to assume that future assets will take losses in a similar manner.
Third-party timing curves, such as those produced by the Public Securities Associa-
tion (PSA) or rating agencies can be used to adjust losses. Also, more sophisticated
statistical analyses can be performed on the loss data to determine trends. The results
of such analyses would provide a basis for trending. The continuation of Model
Builder 4.2 takes the most fundamental approach to projecting loss.

MODEL BUILDER 4.2 CONTINUED

1. The final step in a complete static loss analysis is adjusting newer vintages to
account for trending. To do this, the monthly loss for each vintage that is not
complete needs to be extrapolated based on timing. First, make room to work
underneath the monthly loss percentage area. Insert enough rows so rows 64
through 67 are clear.

2. Label row 64 in B64 as Loss Sev. Taken. This is how much loss as a percent
of original balance has been taken for each vintage. To get the correct amount
a SUM formula with the OFFSET function needs to be used. For the OFFSET
to reference the correct amount of information per vintage create a row of
descending values starting with 24 in C36, 23 in D36, and so on. Descend the
values until Z36 where the value should be 1. In C64 enter the following formula:

=SUM(C39:OFFSET(C38,C36,0))

This formula will only sum the severities that are derived from historical data.
The importance of the OFFSET becomes clearer later as projected severities are
created in the area.

3. In the next row down, label cell B65 Loss % Taken. This row is a percentage
calculation of how much loss the vintage under analysis has taken compared to
the weighted average timing curve. For instance, the January 2004 vintage has
a full 24 months reported, so it has taken 100 percent of loss that it is expected.
The February 2004 vintage is only 23 months so it is short one month of loss
and has taken slightly less than 100 percent loss. To calculate the percentage of
loss that has been taken, enter the following formula starting in C65:

=OFFSET($AE$38,C36,0)

This formula is a basic OFFSET for the timing curve, depending on the seasoning
of the vintage. Copy this formula through to Z65.

4. By knowing the percentage of loss that has been taken, the calculation for the
percentage of loss that needs to be distributed is determined by subtracting the
prior value by one. Label cell B66 Loss to be Dist. Enter the following formula
in cell C66 and copy it across to Z66:

=1−C65

5. The expected loss is the loss severity taken divided by the loss percent taken
so far. If a vintage has taken 100 percent of its loss, then it will be the same
loss severity, however for vintages that have taken less than 100 percent the
severity will be grossed up. Label cell B67 Expected Loss and enter the following
formula in C67 and copy it across to Z67:

=C64/C65

6. With the expected loss for each vintage calculated, the next step is to project the
monthly loss for periods in the future. This can be done directly in the monthly
loss formula since there is already an IF statement set up. Click on cell C39 and
recall that an IF statement was set so that if there was no data (that is ‘‘’’), then
no data should be populated. However, it is now known that if there is no data,
there should be a projection. The projection is going to be the expected loss
amount multiplied by the projected timing of loss. This is summarized by the
following formula that cell C39 should be updated to:

=IF(C8="",C$67*$AD39,C8/C$38)

This formula reads: If there is no monthly loss data project it by taking a
projected timing curve and multiplying that curve by the expected loss amount,
otherwise the loss is based on historical data. Copy this formula across the range
C39:W62. Only this range should be used since October 2005 onwards has so
few data points that the calculations will cause #DIV/0 errors. At this point the
bottom part of the monthly loss table should look like Figure 4.10.

FIGURE 4.10 The additional rows are used to project expected loss.

7. The last step is to create a new weighted average curve, taking into account
the projected amounts. Label cell AG38 Adj. WA Curve and in AG39 enter the
following formula:

=SUMPRODUCT(C39:W39,$C$38:$W$38)/SUM($C$38:$W$38)

Copy this formula down to cell AG62. This is a straightforward weighted
average formula, taking into account ALL of the data for each period (up to
column W). When the individual monthly data is summed in cell AG69, the
difference is apparent between using an adjusted curve and a purely historical
curve when trending is taking place. In the latter example, a loss curve of 9.34
percent would be used, while in the former case a much lower loss curve of 7.01
percent would be used due to trending. See Figure 4.11 for a comparison.

The previous sections described in detail the most common analyses performed
on static loss data, however it is by no means exhaustive. There are many sit-
uations that will require different methodologies such as extremely volatile data,
an insufficient quantity of data, a change in assets, etc. Understanding the fine
detail of each situation and what drives loss is the key to choosing the right
methodology. Two different static loss histories may appear very similar, but the
methodology that should be employed often depends on information that is not
on the data tape. These other methodologies can range from calculation inten-
sive analysis, such as examining the slopes of the worst vintages to a very simple
comparables study.

Regardless of the methodology that is used to analyze loss, understanding loss
and what causes it in a transaction is possibly the most important component of
structured finance modeling. A majority of the structure revolves around the loss
and exists to mitigate it. This will become more apparent as loss expectations are
implemented in the model.

INTEGRATING LOSS PROJECTIONS

The first part of this chapter focuses on understanding loss from a historical
perspective and attempting to extrapolate future loss from the history. This second
part takes the knowledge garnered from the history and applies it so loss can be

FIGURE 4.11 The new adjusted
weighted average curve is less than the
original weighted average curve after
a decreasing loss trend is taken into
account.

taken into account when generating cash flows. Two methods of calculating loss exist
for structured finance modeling: original balance calculation and current balance
calculation. The correct one to use depends on the type of loss curve that is integrated
into the model.

The first method, original balance calculation, multiplies a monthly loss severity
by the original balance of the assets. This is used when historical loss analysis has
been completed on assets and when historical loss severities have been calculated off
of original balance. If 100 percent of the timing curve is taken and there is no credit
for seasoning, the dollar loss amount as a percentage of the asset original balance
should be exactly the same as the gross cumulative loss assumption.

The other method calculates loss by multiplying a monthly default rate (MDR)
by the current balance. Monthly default rates are primarily employed when using
a Standard Default Assumption (SDA) curve as the loss projection. In this case the
dollar amount of loss will not be related to a percent of the original balance.

Regardless of the methodology, something to realize about loss projection is that
it is a percentage of the asset balance. This does not seem that unusual when using a
representative line style of amortization. The assets have been aggregated and should

therefore have percentages of loss taken out. However, it may seem unusual when
using a loan level style of amortization because a percentage of loss is taken out of
an individual loan. In reality a loan will either default or not. There is no concept
of part of a loan defaulting. In modeling, however, a loss curve will be applied to
each loan and the results aggregated. This concept becomes more important when
thinking about seasoning and default timing.

The Effects of Seasoning and Default Timing

When a loan has begun to amortize or is seasoned, the expected loss amount will
change because a seasoned loan is on a different part of the loss curve than a new
loan. For example, a loan that is brand new with a final maturity of 24 months
might have a loss curve that is 24 periods in length. By month 24 the loan will have
taken 100 percent of its expected loss. Imagine that the loan was already 10 months
old when it was sold into the transaction. This means that 10 months of loss should
be expected to already have taken place. Figure 4.12 shows the difference of two
loans with different seasoning and their expected remaining loss.

FIGURE 4.12 A new loan will be expected to
take a full 7.01 percent of loss, while a loan
seasoned 10 months is assumed to have already
taken 2.31 percent loss, leaving the expectation
of 4.70 percent of loss to be incurred.

The effects of seasoning are accounted for in a model by calculating the seasoning
of a representative line or individual loan and making sure that the loss applied for
each period corresponds to the correct place on the default curve.

Seasoned loans can also have very different loss expectations depending on the
default timing curve. Earlier, default timing and the problems that can arise from
different default timing curves was discussed. However, all of that analysis assumed
a new loan. If a loan is seasoned and the default timing curve is front loaded, there
is a good chance that the loan has already taken a significant amount of its expected
loss. Once Project Model Builder is complete, the differences in loss expectation due
to seasoning and default timing can be examined by varying the loan age and timing
curve.

MODEL BUILDER 4.3: INTEGRATING DEFAULTS IN ASSET AMORTIZATION

1. Start on the Inputs sheet and label the following cells:
E17: Gross Cumulative Loss
F17: Loss Stress
G17: Loss Timing Curve
H17: SDA Curves
Underneath each label is where the values will be entered. For now enter 1.00%
in cell E18 and name this cell pdrCumLoss1, enter 1 in F18 and name the cell
pdrLossStress1. Before cells G18 and H18 can be created, some work needs to
be done on the Vectors sheet.

2. On the Vectors sheet Chapter 3 ended on column R. Leave column S blank for
spacing purposes and label cell T4 Defaults. Columns T through X are where the
timing curves will be stored. Label cells T5 through X5 Timing Curve 1, Timing
Curve 2, and so on. Name the range S5:X5 lstDefaultCurve. It is important to
include the blank cell S5 so the data validation list will have the option of a
blank value.

3. While on the Vectors sheet move on to cells Z5:AD5. Label these cells Default
Rate 1, Default Rate 2, and so on. Make sure to leave Y5 and AE5 blank for
spacing purposes. Move on to cell AF5 and label that cell SDA 50%, AG5 SDA
100%, and AH5 SDA 200 %. Name the range AF5:AI5 lstSDA.

4. Go back to the Inputs sheet and create a data validation list in cell G18 using
lstDefaultCurve as the list range. Name cell G18 pdrLossTime1. Create another
data validation list in H18 using lstSDA. Name cell H18 SDA Loss.

5. At this point there is an input for the loss severity and a selector for timing.
The severity can be entered and changed quickly depending on the historical
loss analysis results. The timing curve has been set up so there are five curves to
choose from. Up to this point only the labeling has been created, so an actual
system of determining timing needs to be implemented. This is best done with a
table that allows time to be parsed in a flexible manner, with the timing of loss
varying between time increments. Since this table takes up room and is different

from most of the other items in the model, insert a new sheet after the Cash
Flow sheet and name it Loss Timing. On the Loss Timing sheet, label cell A4
Loss Timing. Label cell A6 Months. Cells D6 through H6 will be the labels for
the loss timing curves. Use a numbering system from 1 to 5, 1 being the number
entered for cell D6, 2 for E6, and so on. At this point, the sheet should look like
Figure 4.13.

6. Still on the Loss Timing sheet enter a 1 in cell A7. This represents the first period
that the loss timing starts with. In cell B7 enter 12. This represents period 12 on
the loss timing curve. What is being created here is the parsing of time that will
be referenced later; in this case period 1 through period 12. A quick method
of making this appear as a label, but retain the number values for referencing
purposes later is to use a custom format for the cell. Right-click cell A7 and
click Format Cells. In the Format Cells dialog box, click the Number tab, select
Custom as the category. In the Type text box enter #,## ‘‘to’’. This should make
the cell look like the cell in Figure 4.14.

The cell will still have a numerical value, but can be read quickly as a parsing
of time. The cells below A7 and B7 should increase according to the interval of

FIGURE 4.13 The loss timing sheet is structured so loss scenarios can be
toggled quickly.

FIGURE 4.14 Using a custom cell format retains the numerical value creating
greater functionality for references later.

time. In this case, cells A8 and B8 will be 13 and 24 respectively. Continue this
pattern down through row 36 so there is a maximum of 360 periods.

7. The purpose of the table made in step 6 is to create possible loss timing scenarios.
Scenario 1 (labeled so in cell D6) will have percentages in cells D8 through D36
that represent the timing of loss during each interval that was set up in the A and
B columns. For example, enter 3.33333333%—or simply enter = 100/30 as an
easier way to get this value—in cell D8. This means that 3.33333333 percent
of the loss severity will be applied to assets in the first year of their term. For
instance, if the loss severity over the life of an asset is expected to be 10 percent,
.33333333 percent (10% * 3.33%) would be expected to occur in the first 12
months. For now assume that 3.33333333 percent of loss will occur in each
interval for Scenario 1 (D8:D36). For 360 periods parsed equally into years this
should equal 100 percent. In fact, a complete timing curve should always equal
100 percent, otherwise an incorrect loss amount is being applied. The other loss
timing scenarios can be left blank for now. Later in the book, when scenario
selection is explained, the other timing scenarios will be entered.

8. Loss timing is often expressed as intervals of time (such as 3.33333333 percent
in months 1 to 12), but models are typically run more granularly such as
monthly, therefore loss timing needs to be converted to the model’s periodicity.
Ultimately a monthly vector will be created so the most logical place to store this
vector is on the Vectors sheet. Remember that in step 2 an area was created for
five Timing Curves (columns T through X). An OFFSET-MATCH combination
is the formula that will be used to pull the correct periodic loss timing. In cell
T7 on the Vectors sheet, enter the following formula:

=OFFSET('Loss Timing'!D$6,MATCH($A7,'Loss Timing'!$A$7:

$A$36,1),0)/12*PmtFreqAdd

This formula is similar to the others that use OFFSET-MATCH, with a few
exceptions. In this case the start of each loss timing scenario is referenced by
column (D$6). That reference cell is offset by matching the current period on the
Vector sheet against the intervals in column A on the Loss Timing sheet only.
The fact that column A is only used is extremely important for this formula
to work correctly. The reason this column is only used is because the type of
MATCH that is being used is set to a 1. This means that the formula will find
the largest value that is less than or equal to the look up value. If the rate for
period 14 were trying to be determined, the largest value on the Loss Timing
sheet’s cells A7:A36 is 13. This corresponds to the second interval of timing on
the Loss Timing sheet, which is the correct interval to be referenced (13 to 24).
A 1 match type works only in the case of referencing the lower bound of the
intervals.

The other exceptions are the divisors in the formula. The amount returned
from the OFFSET-MATCH is based on the interval. To get to the periodic
amount the interval amount needs to be divided by the model’s periodicity. If

FIGURE 4.15 The timing curve is represented on
a monthly basis on the Vectors sheet.

the model was always monthly then all that needs to be done is to divide by 12.
However, to automate the model in case the periodicity is quarterly, semiannual,
or annual multiplying by the Payment Frequency Additive is necessary. Make
sure to copy the completed formula through T366. So far this area should look
like Figure 4.15.

9. Still on the Vectors sheet, the next step is to come up with the correct periodic
default rate. This is the final rate that will be applied to a balance to come up
with a dollar amount of loss. This rate consists of severity multiplied against
periodic timing. Also, this area is where any stress should be applied to the loss
curve. Recall that in Step 3 columns Z:AD on the Vector sheet were set aside
for this purpose. In Z7 enter the following formula:

=(pdrCumLoss1*pdrLossStress1)*T7

The formula takes the overall loss severity from the Inputs sheet (pdrCumLoss1),
multiplies it by a stress factor if desired (pdrLossStress1), and then multiplies
that product by the current period’s timing. This formula will produce the rate
that should be applied against the dollar balance to derive the dollar loss amount
for a period. Copy this formula into the range Z7:AD366.

10. So far this section has focused on user-generated loss curves; however, there
are times when a preexisting loss curve should be used, particularly with
long-term assets such as mortgages. Earlier an area was set aside for Standard
Default Assumption (SDA) curves. These curves are fixed amounts that have
been determined by the Public Securities Association (PSA) using decades of
historical data from the U.S. mortgage market. They serve as excellent proxies
to determine loss for mortgage products and occasionally other long term assets.

The most basic SDA curve is 100 percent SDA, which assumes an increase
of .02 percent annual default in the first 29 months (starting with .02 percent),

FIGURE 4.16 100 percent SDA displayed as a line graph.

a level .60 percent annual default for months 30 to 60, and then a decrease
of .0095 percent annual default for months 61 to 120, and finally a level .03
percent for months 121 through 360. 100 percent SDA has a very recognizable
shape in the mortgage industry when graphed as in Figure 4.16.

Multiplying or dividing the values of the 100 percent curve creates variations
of the curve. So a 50 percent SDA curve contains half of the values for each
period of the 100 percent curve, while a 200 percent SDA curve contains twice
the values for each period of the 100 percent curve.

It is important to note that the values from these SDA curves are not the
ones used in a monthly model. Remember that SDA uses a monthly default
rate and the curve constructed above is created with annual rates. Use the
following formula to convert from an annual default rate to a monthly default
rate:

Monthly Default Rate = 100∗(1−(1−(Annual Default Rate/100))(1/12))

The values for 50 percent, 100 percent, and 200 percent SDA are stored in
the completed model in the Vectors sheet (AF7:AH366). Copy and paste these
values into the same section of the model being created.

11. The next step that starts bringing the assumptions together is on the Cash Flow
sheet. Go to the Cash Flow sheet and recall that columns M and N were created
and labeled for default information. Cell M7 needs to contain the correct default
rate for the asset pool depending on the selections from the Inputs sheet. Since
there are two types of loss curves that can be used, user generated or SDA, the
formula to determine the default rate will need to have a function that selects
the correct curve based on the Inputs sheet.

One method to select the correct curve is to check to see if one of the Input
sheet cells that select the curve is blank. If a user selected no value for one of
the curves, the other curve must be the one being used. To make sure this is

possible, blank cells were included in the ranges as part of the data validation
lists. To make this clearer, the beginning of the formula that should be entered
in M7 starts with an IF statement involving one of the curve assumptions on the
Inputs sheet:

=IF(pdrLossTime1="",

The IF statement checks to see if the Timing Curve assumption is blank. If
this is true, then an SDA curve must be in use; however, if it is false then a
user-generated curve must be in use. It is important as a model operator to
always keep one of the values blank, otherwise there can be confusion and
formulas may not work as intended. The completed formula should appear as:

=IF(pdrLossTime1="",OFFSET(Vectors!$AE$6,A7+Age1,

MATCH(SDA Loss,lstSDA,0)),OFFSET(Vectors!$X$6,A7+Age1,

MATCH(pdrLossTime1,lstDefaultCurve,0)))

OFFSET-MATCH combinations are used to look up the correct default rate
from the Vectors sheet depending on the type of curve and the age-adjusted
period. Copy and paste this cell down to M366.

12. The final cell to complete the default section is the dollar value calculation of the
periodic defaults in N7. This appears to be a simple formula involving default
rate multiplied by balance, but involves a couple of nuances depending on the
type of default rate being used. A user-generated default curve is calculated
against the original balance of each vintage. If such a curve is being used as
the default assumption, then the default rate should be multiplied against the
original balance of the assets. However, the SDA curve was originally calculated
using the current balance to produce a monthly default rate. If an SDA curve is
being used then the default rate should be multiplied against the current balance
of the assets. The necessary toggle between the user-generated and SDA curve
assumption is implemented with an IF statement similar to step 11:

IF(pdrLossTime1="",L7*M7,$L$7*M7)

If an SDA curve is being used then the current balance is multiplied by the
default rate. If a user-generated curve is being used then the original balance
is multiplied by the default rate. In the case of basing defaults against original
balance, there is a chance that the calculated periodic default could be higher
than the current principal balance. Since it is impossible to default more asset
than exists, this needs to be cleaned up using a MIN formula:

=MIN(L7,IF(pdrLossTime1="",L7*M7,$L$7*M7))

Copy and paste this formula over the range N7:N366.

13. Two additional concepts on the Cash Flow sheet can be finished off once
the default amount is known: actual amortization and actual interest. First,
actual amortization is completed in cell R7. The proper amount of principal
is calculated by taking the beginning balance of the assets less the defaulted
amount and multiplying that by a ratio that represents the notional amortization
for the period. Entering the following formula in R7 accomplishes this:

=(L7−N7)*(1−O7/O6)

Notice that the notional amortization ratio is the current period’s amortization
factor divided by the prior period’s. Copy and paste this formula over the range
R7:R366.

14. Finally this section is completed by entering formulas in two columns for the
actual interest calculation. The first formula carriers over the interest rate from
the notional schedule. Enter the following in cell S7 and copy and paste it over
the range S7:S366:

=F7

Next enter the following formula in T7:

=S7*C7*(L7−N7)

This calculates the dollar amount of interest without the defaulted loans. Copy
and paste this formula over the range T7:T366.

With loss calculations complete, it is important to take a step back and
understand how the loss affects the model in general. Defaults are the principal
balance of loans that are assumed to be losses. Since this principal amount is
uncollectible, it should be removed from the asset pool balance, which is done in
the Ending Balance formula in column V of the Cash Flow sheet. It is critical to
realize that the asset pool balance is reduced each period by defaults. The actual
amortization and interest are two concepts that are immediately and most noticeably
affected by defaults. Many other concepts later in this text are affected by loss so its
calculation and application should be well understood.

The importance of analyzing delinquency, default, and loss needs to be reiterated.
Loss is a core determinant of the performance of a transaction and should be
examined carefully. Such analysis is particularly critical for correctly sizing risk-rated
tranches and understanding the risk of owning an asset that has the possibility of
loss. While this chapter is an excellent introduction to these topics, asset specific
characteristics could require additional analysis and unique methodologies.

Prior to moving on to the liability side of the model, there is one more section
related to the assets: recoveries. After a default occurs, there is the possibility that the
repossession and sale of the defaulted asset and the continued pursuance of credit
collections will recover cash. This concept is the focus of Chapter 5.


